

# The Hodgkin-Huxley Nerve Axon Model

Gary D. Knott  
Civilized Software Inc.  
12109 Heritage Park Circle  
Silver Spring MD 20906  
phone:301-962-3711  
email:knott@civilized.com  
URL:www.civilized.com

May 27, 2015

## 0.1 The Hodgkin-Huxley Model

This technical note shows the use of the *MLAB* mathematical and statistical modelling system for solving the *Hodgkin-Huxley* differential equations for arbitrary initial conditions.

The prevailing model of a nerve axon membrane is a pair of theories concerning the nature of the axon membrane with respect to active (“pumping”) and passive (diffusion) steady-state transport of various ions across the membrane and with respect to time-dependent “gate” opening and closing which controls the active passage of ions through such “open gates”.

It is postulated that the membrane in a given state has a certain permeability for each given ion, and that this membrane state is determined by the electrochemical potential across the membrane.

The permeability,  $P_C$ , of a membrane for a particular chemical species,  $C$ , is a measure of the ease of diffusion of  $C$  across the membrane in the presence of a concentration difference on either side of the membrane. In particular, upon entering the membrane, one  $C$ -molecule will travel, on the average,  $(P_C a F / (\beta R T)) E_m$  cm/sec transversely across the membrane, where  $P_C = u \beta R T / (a F)$ , and  $E_m$ ,  $u$ ,  $\beta$ ,  $R$ ,  $T$ ,  $a$ , and  $F$  are defined as follows.

$E_m$  = the electric potential difference across the membrane, measured in volts.

$u$  = the electric mobility of  $C$  in the membrane  $\approx .001$  (cm/sec)/(volt/cm) for sodium or potassium. This is the average velocity of a  $C$ -molecule due to an electric potential difference of one volt/cm (e.g., for a 100 Angstrom membrane, the reference potential difference is  $1 \mu$ volt between the inside and the outside). As the potential difference,  $E_m$ , changes, the mobility of  $C$  changes. For a  $-60$  mv potential difference across a 100 Angstrom membrane,  $u \approx 600$  (cm/sec)/(volt/cm).

$\beta$  = the partition coefficient between the membrane and solution  $\approx .003$ .  $\beta$  is the ratio of the concentration of  $C$  in the solution to the concentration of  $C$  in the membrane itself in steady-state.  $\beta$  is a measure of the relative affinity of the membrane substance for  $C$ -molecules with respect to the neighboring solution.

$a$  = the thickness of the membrane, about 100 Angstroms.

$F$  = Faraday's constant, 96486.7 coulomb/mole.  $F$  is Avogadro's number times the charge in coulombs of a proton.

$R$  = the gas constant, 8.31434 joule/(degree/mole).  $R$  is Avogadro's number times Boltzmann's constant.

$T$  = temperature in degrees Kelvin.

For a given ionic species,  $C$ , which is placed in unequal concentrations on either side of an idealized membrane, there will be a net diffusion of  $C$  in the direction of lower concentration. This diffusion changes any existing electric potential field, since charge is being redistributed. Other species may be diffusing as well, but regardless of the total complexity, the ionic current of  $C$  ions will be zero just when the electric potential difference across the membrane is sufficient to induce a compensating counter drift of  $C$ -ions in the direction of lower concentration. The potential difference of the potential in the "in"-compartment minus the potential in the "out"-compartment which leads to a net  $C$ -ion current of zero is given by the so-called Nernst potential difference,  $E_C = (ZRT/F) \log(C_{out}/C_{in})$  volts, where  $Z$  is the valence of  $C$  (+1 for sodium and  $-1$  for chlorine) and  $C_{in}$  and  $C_{out}$  are the concentrations of  $C$  in the inside and outside compartments. Note when  $C_{out} = C_{in}$ ,  $E_C = 0$ . The Nernst potential difference is essentially obtained by converting the concentration difference  $C_{out} - C_{in}$  from Moles/liter units to Volts/liter units using knowledge of Faraday's constant.

The actual electrical potential across the membrane is a function of the relative concentrations of each species present, and it is this total potential, together with its diffusion tendency, which determines how each ion would move. The total potential can be computed using the Goldman equation given below. Potential is defined so that positive charge flows towards points of lower potential, while negative charge flows towards points of higher potential. The membrane potential difference at a point is defined as the potential at the point minus the potential at a reference point in the outside solution.

For any given initial concentrations of various species and given time-course changes of membrane permeabilities for these species, there will be an associated change in total potential and a varying ionic current flow which, in principle, we can determine with appropriate equations. (We ignore the added complication of concentrations changing due to the diffusion of the solute, or of hydrostatic pressure gradients arising when solute diffusion is not possible.)

A full discussion of nerve membranes is given in "Ionic channels of excitable membranes" by Bertil Hille (Sinauer Associates, Sunderland, Mass, 1984).

For a giant squid nerve axon it has been experimentally determined that the chemical species of interest are potassium ions, sodium ions, chloride ions, and various other ions of lesser importance.

Ions of all these less-significant species are collectively lumped together and called “leakage” ions. Each of the various leakage ions have differing permeabilities; the result is that the permeability for leakage ions is an “average” which compensates for the various differing permeabilities actually involved. Moreover, some of the leakage ions are negative and some are positive. We shall assume that our “average” leakage ion is negative, and that the strength of an ionic current of leakage ions is diminished to compensate for this fiction. The resting axon membrane of an “average” squid has permeabilities in cm/sec units for these species which are given in the table below with the resting concentrations in mmoles/liter.

	inside	outside	resting permeability	potential difference
$K(+)$	400	10	$6 \times 10^{-6}$	-72 mV
$Na(+)$	50	460	$8 \times 10^{-9}$	55 mV
leakage(-)	45	540		-50 mV

These values lead to a zero net ionic current in the resting state. The total resting state potential difference,  $E_r$ , based on the Goldman equation works out to -60 mv. The Goldman equation in this case is:

$$E_r = \log((P_K K_{out} + P_{Na} Na_{out} + P_L L_{in}) / (P_K K_{in} + P_{Na} Na_{in} + P_L L_{out})) RT / F$$

volts, where  $L$  denotes “leakage” ions. This is valid only when there is zero net ionic current. Here  $K_{in}$  and  $K_{out}$  denote the concentration of potassium in the “in” and “out” compartments,  $Na_{in}$  and  $Na_{out}$  denote the concentration of sodium in the “in” and “out” compartments, and  $L_{in}$  and  $L_{out}$  denote the concentration of leakage ions in the “in” and “out” compartments. The coefficients  $P_K$ ,  $P_{na}$ , and  $P_L$  are the membrane permeabilities of potassium, sodium, and leakage ions respectively. Note the “in” and “out” concentrations of the leakage ions are “reversed” because the leakage ions have negative valence.

The above state cannot be indefinitely maintained by a passive membrane which allows each species to pass at a non-zero rate. Even though the net ionic current is zero, there are non-zero sodium, potassium, and leakage currents in the resting state. An additional mechanism is postulated, namely a “sodium pump” which actively transports sodium ions from the inside to the outside to balance the resting state inward-diffusion, and a “potassium pump” which transports potassium ions from the outside to the inside to balance the resting state outward-diffusion.

There is certainly such mechanisms present in the axon membrane but their precise chemical natures are not completely known. The primary pump is the sodium-potassium ATPase pump that brings two potassium ions inside for every three sodium ions that are pumped-out. The energy to drive this pump depends upon the conversion of  $ATP$  to  $ADP$ . A discussion of current research results about the pumping mechanism can be found in “The  $(Na^+ + K^+)$  activated enzyme system and its relationship to transport of Sodium and Potassium” by Jens Chr. Skou, pp. 401:434 of Quarterly Reviews of Biophysics, Vol. 7, No. 3, July 1974.

Conductance is merely the reciprocal of resistance. It is a measure of the ability of a material to carry a particular ionic or electric current. Conductance is measured in  $\Omega^{-1}/\text{cm}^2$  units. The potassium conductance of one  $\text{cm}^2$  of axon membrane is denoted by  $g_K$ , The sodium conductance

of one  $\text{cm}^2$  of axon membrane is denoted by  $g_{Na}$ , and the leakage ion conductance of one  $\text{cm}^2$  of axon membrane is denoted by  $g_L$ . These values are, in general, functions of the potential difference history of the membrane.

Ohm's law holds for ionic current flow, but since there is zero net current for the ionic species,  $C$ , when the voltage is exactly the Nernst potential,  $E_C$ , for  $C$ , we have the  $C$ -ion current is given by  $I_C = g_C(E_m - E_C)$ , where  $E_m$  is the total membrane potential. This means  $I_C$  is 0 when  $E_m = E_C$ , and in any case, the  $C$ -ionic current is  $g_C(E_r - E_C)$  in the resting state. Ohm's law applies here by convention, since  $g_C$  is determined in any state, just so  $I_C = g_C(E_m - E_C)$  is true. Here  $C$  is either  $K$  (potassium),  $Na$  (sodium), or  $L$  (leakage ions).

The conductance,  $g_C$ , and the permeability,  $P_C$ , for a given ionic species,  $C$ , with charge  $Z$ , are related as follows.


$$P_C = g_C RT (E_m - E_C) \cdot (\exp(Z E_m F / (RT)) - 1) / ((C_{out} - C_{in} \exp(Z E_m F / (RT))) / Z / (F^2 E_m)),$$

for  $C_{out} - C_{in} \exp(Z F E_m / (RT)) \neq 0$ , where  $E_m$ ,  $E_C$ ,  $F$ ,  $R$ ,  $T$ ,  $C_{out}$ , and  $C_{in}$  are as defined above.  $E_m$ ,  $E_r$  and  $E_C$  are here measured in volts.

The Hodgkin-Huxley model of an axon membrane is a mathematical description of the potential difference or voltage across the membrane which changes as a function of time in response to various perturbations of potential established with an associated applied current. It was presented in the Journal of Physiology, Vol. 117, pp. 500:544, 1952.

This description is in the form of an analogous circuit which in turn can be characterized by a set of non-linear differential equations. These equations contain an empirically adequate description of the feedback system which changes the conductances (or equivalently, the permeabilities) of the membrane as a function of time and potential difference.

The circuit merely contains the three main ionic current circuits in parallel together with the observed membrane capacitance.



Empirically we have the emf values  $E_K \approx -72$  mV,  $E_{Na} \approx 55$  mV, and  $E_L \approx -50$  mV.

The values  $g_K$ ,  $g_{Na}$ , and  $g_L$  are the conductances of the membrane for potassium, sodium, and leakage ions respectively.  $g_K$  and  $g_{Na}$  are functions of time and voltage.

The sign convention for current is such that a net flow of positive ions from inside to outside is a positive current. This is the “electron” convention as opposed to the conventional Franklin convention. Thus a net flow of positive ions from outside to inside, or a net flow of negative ions from inside to outside is a negative current. Here current is measured in  $\mu$ Amperes per  $\text{cm}^2$ , and  $E_m$ ,  $E_K$ ,  $E_{Na}$ , and  $E_L$  are measured in mvolts.

The values of  $E_K$ ,  $E_{Na}$ , and  $E_L$  given above imply that independently in the resting state, we would have a nearly-zero sodium current, a nearly-zero potassium current, and a zero leakage ion current. The sodium and potassium currents just balance the corresponding pump currents.

These circuits are connected in parallel however, and thus the total driving potential applied to sodium ions is  $E_m - E_{Na}$ , the total driving potential applied to potassium ions is  $E_m - E_K$ , and the total driving potential applied to leakage ions is  $E_m - E_L$ . According to Kirchoff’s law, the currents into any point must sum to zero, so these three ionic currents must sum to 0. From Ohm’s law as discussed above, the sodium current,  $I_{Na}$ , is  $g_{Na}(E_m - E_{Na})$ , the potassium current,  $I_K$ , is  $g_K(E_m - E_K)$ , and the leakage ion current,  $I_L$ , is  $g_L(E_m - E_L)$ . But, in the resting state,  $I_K + I_{Na} + I_L = 0$ , so  $E_m = (g_{Na}E_{Na} + g_K E_K + g_L E_L)/(g_{Na} + g_K + g_L)$ . In the resting state,  $E_m$  has been determined experimentally to be about  $-60$  mV,  $g_{Na}$  has been determined to be about .0106092 and  $g_K$  is about .3666445, so we may define

$$g_L = (g_{Na}(E_{Na} + 60) + g_K(E_K + 60))/(-60 - E_L) \approx .3179676.$$

(Usually,  $g_L$  is set to .3, and  $E_L$  is computed to yield a zero current, but we follow the opposite convention.)


Thus, for our “average” axon, the total sodium driving potential is  $-115$  mV and we have a negative sodium current, the total potassium driving potential is  $12$  mV and we have a positive potassium current, and the total leakage ion driving potential is  $-10$  mV and we have a negative leakage ion current.

The sodium current is approximately  $-1.22$  mA, the potassium current is about  $4.4$  mA, and the leakage ion current is about  $-3.18$  mA, and, of course, the sum is zero.

Of course, as these currents flow, the concentration-based emf’s  $E_K$ ,  $E_{Na}$ , and  $E_L$  change; however the pumping devices postulated before are assumed to maintain a  $1.22 \mu\text{A}$  sodium current and a  $-4.4 \mu\text{A}$  potassium current which forces leakage ions to remain in place in order to keep charge balanced in both the inside and the outside compartments. Thus in the resting state there is zero net ionic current across the membrane, and zero individual ionic currents as well. The Hodgkin-Huxley model does not include the pump current, so although there is zero net current in the resting state, the individual ionic currents are presented as computed above.

We may apply a time-varying current,  $-I_a$ , across the membrane as a stimulus. This can be achieved by the feedback-controlled circuit shown below which applies whatever time-varying voltage,  $V_a$ ,

across the membrane is needed to achieve the current  $-I_a(t)$  at time  $t$ . Here time is measured in milliseconds.



Thus, in general, treating the “inside” as a junction of our five parallel circuits, we have the five currents  $-I_a$ ,  $I_K$ ,  $I_{Na}$ ,  $I_L$ , and  $I_c$ , summing to 0, where  $I_c$  is the current across the membrane due to the capacitance  $c$  and the applied current  $I_a$  is taken oppositely-signed from the other four circuit currents. Altogether we have  $I_a = I_K + I_{Na} + I_L + I_c$ .

Now the voltage drops in the  $[V_a, \text{capacitor}]$  loop sum to zero, so  $-V_a + Q/c = 0$  where  $Q$  is the time-varying charge difference in the “plates” of the membrane capacitor. And we have  $\frac{dQ}{dt} = I_c$ , so  $-\frac{dV_a}{dt} + \frac{I_c}{c} = 0$ . Thus  $-I_c = cV_a'$  and  $V_a$  is just the total membrane potential difference  $E_m$ ! And then  $I_c = c(dE_m/dt)$ , where  $c \approx 1 \mu\text{farad}/\text{cm}^2$ . Thus, we obtain

$$c(dE_m/dt) = I_a - g_K(E_m - E_K) - g_{Na}(E_m - E_{Na}) - g_L(E_m - E_L).$$

The Hodgkin-Huxley model assumes that the emf's  $E_K$ ,  $E_{Na}$ , and  $E_L$ , and the conductance  $g_L$ , are constant, but that the conductances  $g_K$  and  $g_{Na}$  are not constant, but rather are functions of the membrane potential difference history.

A change in  $g_K$  corresponds to a change in the permeability of the membrane for potassium and similarly for  $g_{Na}$ . The physical idea stated below underlying the Hodgkin-Huxley specification of how  $g_K$  and  $g_{Na}$  change is, by itself, incorrect, (see Armstrong, *C.*, *Physiological Reviews*, Vol. 61, pp. 644:683, 1981). However, the associated equations adequately serve the purpose of defining  $g_K$  and  $g_{Na}$  as functions of time and membrane potential difference.

The Hodgkin-Huxley definition of  $g_K$  and  $g_{Na}$  can be “explained” as follows.

Let the opening of a potassium “channel” which allows potassium to pass through the membrane involve  $j$  events. When  $j = 4$ , for example, we have (inside)  $K \Leftrightarrow K' \Leftrightarrow K'' \Leftrightarrow K''' \Leftrightarrow K''''$  (outside), and let us imagine that each such event is enabled by a  $K$ -gate molecule in the on-state. In other words, we imagine a potassium channel is composed of  $j$  gate-molecules, all of which must be in the on-state for the channel to be open. If  $n(t, E)$  is the proportion of  $K$ -gate molecules which are in the on-state at time  $t$  and membrane potential difference  $E$ , then  $n^j$  is the probability that any given potassium channel is open; this probability is proportional to the probability that a potassium ion approaching the membrane will succeed in passing through. Note that  $j$   $K$ -gate molecules might actually be one  $K$ -gate molecule which assumes  $2^j$  states, only one of which is “open”.

Suppose that when all the  $K$ -gate molecules are in the on-state, the associated potassium conductance of the membrane is  $\bar{g}_K$ . In general, then, the potassium conductance of the membrane at voltage  $E$  and time  $t$  is  $g_K = \bar{g}_K n^j$  where  $0 \leq n \leq 1$ .

Suppose further, that  $K$ -gate molecules switch reversibly between the on-state and off-state following simple first-order kinetics; so that:  $dn/dt = \phi(\alpha_n(1 - n) - \beta_n n)$ , where  $\phi$  is a temperature compensating factor, and where the “rate constants”  $\alpha_n$  and  $\beta_n$  are functions of voltage only! The factor  $\phi = 3^{(T-6.3)/10}$ , where  $T$  is the temperature in degrees Celsius. Hodgkin and Huxley “guessed” that  $\alpha_n(E) = -.01(50 + E)/(e^{-(50+E)/10} - 1)$  and  $\beta_n(E) = .125e^{-(60+E)/80}$  by extensive numerical simulation and curve-fitting. The functions  $\alpha_n$  and  $\beta_n$  are chosen to “work”; their basic form was obtained from theoretical considerations, assuming that the potential across the membrane acts on charged particles (the gate molecules) to move them in and out of the way within the channels in the membrane.

For sodium, suppose there are both  $Na(1)$ -gate molecules and  $Na(2)$ -gate molecules, and let  $m(t, E)$  be the proportion of  $Na(1)$ -gate molecules in the on-state, and let  $h(t, E)$  be the proportion of  $Na(2)$ -gate molecules in the on-state, and finally, suppose opening a sodium channel across the membrane requires  $p$   $Na(1)$ -events and  $q$   $Na(2)$ -events. Thus,  $g_{Na} = \bar{g}_{Na} m^p h^q$ , where  $\bar{g}_{Na}$  is the sodium conductance when all the  $Na(1)$  and  $Na(2)$  gate molecules are in the on-state.

As before suppose that the  $Na(1)$  and  $Na(2)$  gate molecules are switching between their on and off states following first order kinetics, so that  $dm/dt = \phi(\alpha_m(1 - m) - \beta_m m)$ , and  $dh/dt = \phi(\alpha_h(1 - h) - \beta_h h)$ , where again the rate constants  $\alpha_m$ ,  $\beta_m$ ,  $\alpha_h$ , and  $\beta_h$  are functions of the potential difference,  $E$ , which have been empirically chosen as:

$$\begin{aligned}\alpha_m(E) &= -.1(35 + E)/(e^{-(35+E)/10} - 1), \\ \beta_m(E) &= 4e^{-(60+E)/18}, \\ \alpha_h(E) &= .07e^{-(60+E)/20}, \quad \text{and} \\ \beta_h(E) &= 1/(e^{-(30+E)/10} + 1).\end{aligned}$$

The constants  $\bar{g}_K$  and  $\bar{g}_{Na}$  are approximately  $36 \text{ m}\Omega^{-1}/\text{cm}^2$  and  $120 \text{ m}\Omega^{-1}/\text{cm}^2$  respectively.

For a discussion of current knowledge and theory about the nature of the ionic channels, see “Ionic channels and gating currents in excitable membranes” by Werner Ulbricht, pp. 7:31 in Annual Review of Biophysics and Bioengineering, Vol. 6, 1977, and “Kinetics of channel gating in excitable

membranes” by Leon Goldman, pp. 491:526 of Quarterly Reviews of Biophysics, Vol. 9, No. 4, Nov. 1976.

In general we may have an initial state where a “preconditioning” ionic current,  $I_0$ , has been applied for a time sufficient to establish a new steady-state. Since in steady state,  $dE_m/dt = 0$ , for any  $I_0$ -value we have the resulting steady-state voltage,  $E_{ss}$ , given by  $E_{ss} = \text{root}_E(I_K + I_{Na} + I_L - I_0)$ .

Also, initially we assume  $n$ ,  $m$ , and  $h$  are such that  $dn(0)/dt = dm(0)/dt = dh(0)/dt = 0$ , when  $\alpha_n$ ,  $\beta_n$ ,  $\alpha_m$ ,  $\beta_m$ ,  $\alpha_h$ , and  $\beta_h$  are determined by the steady-state voltage,  $E_{ss}$ .

Thus, we have:  $n(0) = n_{ss}(E_{ss})$ ,  $m(0) = m_{ss}(E_{ss})$ , and  $h(0) = h_{ss}(E_{ss})$  where,

$$\begin{aligned} n_{ss}(E) &= \alpha_n(E)/(\alpha_n(E) + \beta_n(E)), \\ m_{ss}(E) &= \alpha_m(E)/(\alpha_m(E) + \beta_m(E)), \quad \text{and} \\ h_{ss}(E) &= \alpha_h(E)/(\alpha_h(E) + \beta_h(E)). \end{aligned}$$

Thus from Ohm’s Law we have the steady-state voltage  $E_{ss}$  that results in a net 0 current with a “current-clamp” of  $I_0 \mu\text{A}$ .

$$E_{ss}(I_0) = \text{root}_E(\bar{g}_{Na}m_{ss}(E)^p h_{ss}(E)^q (E - E_{Na}) + \bar{g}_K n_{ss}(E)^j (E - E_K) + g_L(E - E_L) - I_0),$$

and  $E_m(0) = E_{ss}(I_0)$ .

The Hodgkin-Huxley model has  $j = 4$ ,  $p = 3$ , and  $q = 1$ .

Note that it is possible to choose  $I_0$  so as to have  $E_m(0)$  be any desired steady-state voltage,  $V$ , namely:  $I_0 = \text{root}_I(E_{ss}(I) - V)$ . Such a choice for  $I_0$  represents a so-called voltage-clamp situation.

The stimulus current,  $I_a(t)$ , is a current which varies as desired from the steady-state applied current  $I_0$ .  $I_a(0)$  need not equal  $I_0$ . The interpretation of this is that we have accommodated the axon membrane to the current  $I_0$ , and then instantaneously switched the applied current to be  $I_a(0)$ . Of course the variables,  $n$ ,  $m$ , and  $h$  are accommodated to  $E_{ss}(I_0)$ , not  $E_{ss}(I_a(0))$ , so they will begin to change in response to this new voltage. Slightly more realism can be had by choosing  $I_a(t) = I_0 + (I_1(t) - I_0)(\text{if } t < s \text{ then } (1 - \exp(-kt)) \text{ else } (1 - \exp(-ks)) \exp(-k(t - s)))$ , for some new current function,  $I_1$ , which starts at time 0 and replaces  $I_0$  at some “rise time” rate  $k$ , and then is cutoff at time  $s$  and decays to zero at the rate  $k$ .

Finally, it is possible to apply an impulse current which changes  $E_m(0)$  by some desired amount,  $V_i$ , without allowing time for  $n(0)$ ,  $h(0)$  and  $m(0)$  to change. In this situation, our initial condition becomes  $E_m(0) = E_{ss}(I_0) + V_i$ .

We have now given the complete Hodgkin-Huxley model for the giant squid axon membrane. There are various improvements; notably the Adelman-FitzHugh extension; however we shall suppose here that a “naked” axon having no Schwann cells is being modeled.

The following MLAB do-file, called HHF.DO, can be used to establish the Hodgkin-Huxley equations by typing DO HHF.



```

"file: HHF.DO";
EL=-50; ENA=55; EK=-72;
GKBAR=36; GNABAR=120; GL=.3179676;

FUNCTION PHI(TEMP)=3^((TEMP-6.3)/10);
FUNCTION F(X)=IF ABS(X)<.00002 THEN 1 ELSE X/(EXP(X)-1);
FUNCTION BM(E)=4*EXP((-E-60)/18);
FUNCTION BH(E)=1/(1+EXP(-3-.1*E));
FUNCTION BN(E)=EXP((-E-60)/80)/8;
FUNCTION AM(E)=F((-35-E)/10);
FUNCTION AH(E)=.07*EXP((-E-60)/20);
FUNCTION AN(E)=.1*F((-50-E)/10);
FUNCTION MINF(E)=AM(E)/(AM+BM(E));
FUNCTION HINF(E)=AH(E)/(AH+BH(E));
FUNCTION NINF(E)=AN(E)/(AN+BN(E));

K=25; S=.2;
FUNCTION IA(T)=I0+(I1-I0)*(IF T<S THEN 1-EXP(-K*T) ELSE \
(1-EXP(-K*S))*EXP(-K*(T-S)));
FUNCTION IONIC(E,M,H,N)=GNABAR*M^3*H*(E-ENA)+GKBAR*N^4*(E-EK)+GL*(E-EL);
FUNCTION EM'T(T)=IA(T)-IONIC(EM,M,H,N);
FUNCTION M'T(T)=PHITEMP*(AM(EM)*(1-M)-BM(EM)*M);
FUNCTION H'T(T)=PHITEMP*(AH(EM)*(1-H)-BH(EM)*H);
FUNCTION N'T(T)=PHITEMP*(AN(EM)*(1-N)-BN(EM)*N);
FUNCTION ISS(E)=IONIC(E,MINF(E),HINF(E),NINF(E));
FUNCTION ESS(I)=ROOT(E,-246,830,ISS(E)-I);
"FOR -62<I<32751, ESS(I) LIES BETWEEN -246 AND 830";

TEMP=6.3; PHITEMP=PHI(TEMP);
V0=-60; I0=0; ESS0=ESS(I0);
VI=0; I1=0;

INITIAL EM(0)=ESS0+VI;
INITIAL M(0)=MINF(ESS0);
INITIAL H(0)=HINF(ESS0);
INITIAL N(0)=NINF(ESS0);

```

The following MLAB do-file, called HHDO.DO, can be used to allow the user to establish any desired boundary conditions and then solve the Hodgkin-Huxley equations established with HHF.DO given above.

```

"file: HHDO.DO";
I0=0;
TYPE "
IF AN INITIAL VOLTAGE CLAMP OF X mV (-245<X<-12) IS \

```

```

DESIRED, TYPE: VO=X.
OTHERWISE SET ANY DESIRED PRECONDITIONING uA CURRENT,Y, \
(-62<Y<1100) BY TYPING: IO=Y.
OTHERWISE TYPE A RETURN (VO WILL BE -60 AND IO WILL BE 0).";
DO KLINE;
IF VO NOT=-60 THEN (IO=ISS(VO));
ESSO=ESS(IO);

VI=0;
TYPE "
ENTER INITIAL IMPULSE POTENTIAL CHANGE,X, BY TYPING: VI=X.
OTHERWISE TYPE A RETURN (VI WILL BE 0).";
DO KLINE;

I1=0;
TYPE "
DEFINE AN APPLIED CURRENT FUNCTION,IA, BY TYPING: \
FUNCTION IA(T)=desired expression. \
OTHERWISE, IF DESIRED, MERELY SET AN INITIAL .2 MSEC \
DURATION STEP CURRENT, s, (IN uA) BY TYPING: I1=s.\
OTHERWISE TYPE A RETURN (I1 WILL BE 0).";
DO KLINE;

TV=0:12:.1;
TYPE "
SET the TIME-VECTOR, TV, BY TYPING: TV=0:ft:dt (TV=0:12:.1 BY DEFAULT).";
DO KLINE;

METHOD = GEAR;
Q=INTEGRATE(EM'T,M'T,H'T,N'T,TV);
TYPE "
Q COL 1=TV, COL 2=EM, COL 3=EM', COL 4=M, COL 5=M', \
COL 6=H, COL 7=H', COL 8=N, COL 9=N'.";

```

For example, we can compute the famous "action potential" curve as follows.

```

*DO HHF
*DO HHDO

```

```

IF AN INITIAL VOLTAGE CLAMP OF X mV (-245<X<-12) IS DESIRED, TYPE: VO
= X. OTHERWISE SET ANY DESIRED PRECONDITIONING uA CURRENT,Y,
(-62<Y<1100) BY TYPING: IO = Y. OTHERWISE TYPE A RETURN (VO WILL BE
-60 AND IO WILL BE 0).

```

```

*
```

ENTER INITIAL IMPULSE POTENTIAL CHANGE,X, BY TYPING: VI=X. OTHERWISE  
TYPE A RETURN (VI WILL BE 0).

\*

DEFINE AN APPLIED CURRENT FUNCTION,IA, BY TYPING: FUNCTION  
IA(T)=desired expression. OTHERWISE, IF DESIRED, MERELY SET AN INITIAL  
.2 SECOND DURATION STEP CURRENT,s, (IN  $\mu$ A) BY TYPING: I1 =  
s. OTHERWISE TYPE A RETURN (I1 WILL BE 0).

\*I1 = 50

SET the TIME-VECTOR, TV, BY TYPING: TV = 0:ft:dt (TV=0:12:.1 BY DEFAULT).


\*

Q COL 1=TV,COL 2=EM,COL 3=EM',COL 4=M,COL 5=M',COL 6=H,COL 7=H',COL 8=N,COL 9=N'

\*DRAW Q COL 1:2

\*VIEW

The resulting picture with added titles is reproduced below.



Many familiar observations can be simulated and the model's response can be investigated by solving the Hodgkin-Huxley equations with various initial conditions. See "Computer analyses of the excitable membrane" by T. Hironaka and S. Morimoto in *Computers and Biomedical Research*, Vol. 13, pp. 36:51, 1980, for a number of interesting numerical experiments which can be performed.

One problem with the Hodgkin-Huxley model is that current can flow in each loop of the circuit analog, but this implies sodium and potassium change into each other! This defect is not serious since the “impossible currents” are small, but it is a conceptual error.

One of the main agreements of the Hodgkin-Huxley model with experiment is the suitability of the function  $E_m(t)$  as a solution to the cable equation which involves the velocity of an action potential wave propagating along an axon.

Consider an infinitely-long sheathed cable of diameter  $\delta$  cm with  $r_i$   $\Omega$  resistance for 1 cm. Let the exterior medium have a resistance of  $r_o$   $\Omega$  for a 1 cm distance along the cable. Let  $I_e(x, t)$  be the current in  $\mu A$  applied to the surface of the short band of cable centered at the point at distance  $x$  along the cable and at time  $t$  with an electrode with a feedback voltage control which develops whatever voltage is required to obtain the specified current. The circuit is from ground through the electrode, then through the cable sheathing and, via leakage, back to ground. Let  $I(x, t)$  be the current in  $\mu A$  flowing across the short band of cable sheathing centered at position  $x$ , at time  $t$ .  $I(x, t)$  arises due to capacitance-driven current, leakage, and, for an axon, active forces in the membrane sheathing itself. Finally let  $V(x, t)$  be the voltage across the sheathing (inside potential minus outside potential) at position  $x$  and time  $t$ .

The cable equation relates these functions as:

$$\partial^2 V / \partial x^2 = (r_i + r_o) \pi \delta I - r_o I_e.$$

For a constant sheathing resistance of  $R_m$   $\Omega$  per  $\text{cm}^2$ , we have

$$I = c((\partial V / \partial t) + V / R_m / (\pi \delta^2 / 4)).$$

If our cable is a giant squid axon, however, the membrane resistance varies, and we may define  $I$  by the Hodgkin-Huxley equations as

$$I(x, t) = c(\partial V / \partial t) + g_K(V - E_K) + g_{Na}(V - E_{Na}) + g_L(V - E_L).$$

Thus, we have

$$\partial^2 V / \partial x^2 = (r_i + r_o) \pi \delta [c(\partial V / \partial t) + g_K(V - E_K) + g_{Na}(V - E_{Na}) + g_L(V - E_L)] - r_o I_e,$$

with  $V(x, 0) = -60$  mV for  $-\infty \leq x \leq \infty$ . All the state variables  $m$ ,  $h$ , and  $n$  are now functions of  $x$  and  $t$ , and assumed to be at their normal resting values for all  $x$  at time 0.  $I_e$  is a perturbing current which will cause potential changes which will propagate along the axon according to the cable equation above.

Another way to begin is to assume that  $I_e = 0$ , but that  $V(x, 0)$  is known as an initial condition, where the axon can be assumed to have been charged by a current pulse function which established  $V(x, 0)$  sufficiently rapidly that  $m$ ,  $h$ , and  $n$  have not responded. For example, we may have  $V(x, 0) = V_0 \exp(-x^2/s^2)$  for some spatial decay constants. We have the additional constraint that  $V(x, t) \rightarrow -60 \mu V$  as  $t \rightarrow \infty$ , which will be necessarily satisfied.

The nature of the solution surface  $V$  for the initial condition  $V(x, 0) = V_0 \exp(-x^2/s^2)$  where  $V_0$  is above threshold, is that the curve  $V(0, t)$  is an action potential-like curve which is propagated along the axon (we ignore propagation towards  $x = -\infty$ ) so that approximately the same curve appears as  $V(x_1, t + x_1/\theta)$ .

The parameter  $\theta$  is the propagation velocity. The initial part of the curve  $V(0, t)$  is artificially established by the initial condition and gradually dies out as time progresses, thus we do not have a pure traveling wave. The curve  $V(x_1, t)$  for a large value of  $x_1$  is almost the same shape as  $V(x_1 + \varepsilon, t + (x_1 + \varepsilon)/\theta)$  however.

If we ignore the transient behavior and assume that  $t$  is translated so that  $V(0, t)$  is propagated at velocity  $\theta$  cm/sec in a nearly unchanged form, then we may take  $V(x, t) = V(x - \theta t, 0) = V(0, t - x/\theta)$ . Let  $s = t - x/\theta$ , and let  $U(s) = V(x, t) = V(0, t - x/\theta) = V(0, s)$ . Then  $\partial^2 V/\partial x^2 = (d^2 U/ds^2)/\theta^2$  and  $\partial V/\partial t = dU/ds$ , and we have:

$$d^2 U/ds^2 = \theta^2 (r_i + r_o) \pi \delta [c(dU/ds) + g_K(U - E_K) + g_{Na}(U - E_{Na}) + g_L(U - E_L)].$$

For  $U(0) = -60 \mu V$  and  $dU(0)/ds = 0$ , the solution is 0 but since  $V(x, t) \neq -60$  mV, even for  $t = 0$  and very large  $x$ , when the surface  $V(x, t)$  is not flat at  $-60$  mV, the appropriate initial conditions for the steady-state wave  $U$  is  $U(0) = U_0$  and  $dU(0)/ds = w_0$  where  $U_0$  and  $w_0$  are very small.  $m(0)$ ,  $h(0)$ , and  $n(0)$  may revert to nearly the resting state values. To obtain a physically meaningful solution, the initial values  $U_0$  and  $w_0$  must be nearly correct. Also an acceptable solution must have  $U(s) \rightarrow -60$  mV as  $s \rightarrow \pm\infty$ . This only occurs for certain discrete values of  $\theta$ . The parameter  $\theta$  is thus analogous to an eigenvalue of a linear operator.

It has been shown that, when  $U_0$ ,  $w_0$ , and  $\theta$  are carefully chosen, then  $U(s)$  approximates the almost steady-state traveling wave observed in a real axon, and moreover that  $\theta$  is approximately the propagation velocity of about  $168000/(\pi\delta c(r_i + r_o))$  cm/sec which is actually observed. This fact constitutes one of the corroborations of the Hodgkin-Huxley model for the squid axon membrane.

Since  $\theta$  cannot be precisely represented, the numerically computed solution  $U(s)$  will not approach  $-60$  mV as  $s \rightarrow \pm\infty$ , but rather will approach  $+\infty$  or  $-\infty$ , depending upon the error in  $\theta$  (and  $U_0$  and  $w_0$ ).